Should I pay $11,000 today for a copier or $3,000 a year for 5 years?

The key to answering this question is attributing a value to the annual payments of $3,000 per year. Let’s assume the cost of capital is 12 percent per year.

The PV function returns the value in today’s dollars (or GBP) of a series of future payments, assuming periodic, constant payments and a constant interest rate. A stream of cash flows that includes the same amount of cash outflow (or inflow) each period is called an annuity, and assuming that each period’s interest rate is the same, an annuity can be valued using the PV function. Here’s the syntax for the PV function:

PV(rate,#per,[pmt],[fv],[type])

■ Rate is the interest rate per period. If you borrow money at 6 percent per year and the period is a year, rate equals 0.06. If the period is a month, rate equals 0.06/12, or 0.005.

■ #per is the number of periods in the annuity. In our copier example, #per equals 5. If payments for the copier are made each month for 5 years, #per equals 60. Your rate must be consistent with #per, of course. In other words, if #per implies that a period is a month, you need to use a monthly interest rate. If #per implies that a period is a year, use an annual interest rate.

■ Pmt is the payment made each period. In our example, pmt is $5,000.

A payment is a positive number, while money received is a negative number.

■ Fv is the cash balance (or future value) that you want to have after the last payment is made. In our copier example, fv equals 0. (If we want a $5,000 cash balance after the last payment, fv would equal $5,000.) If fv is omitted, it’s assumed to equal 0.

■ Type is either 0 or 1. Type indicates when payments are made. If type is omitted or equal to 0, payments are made at the end of each period. When type equals 1, payments are made at the beginning of each period.

The solution to the question about the best way to pay for the copier is included in the file PV.xls, shown in Figure 10-1.

[image: image1.jpg]D8 -Q Jf | =PV(annual_rate,Years,Amount_paid_each_year,0,0)

B c [ F G

5

2 PV Function

3 annual rate 012

4 Amount paid each year $ 3,000.00

5 #Years 5

6

i Value of Payments
last 2
arguments

End of Year 510 814 33)|(510.614.33) omitted
9 Beginning of Year (512.172.05)
10

1

extra $500 paid at end of year 5 (511.095.04)




Figure 10-1.

In cell D8, I computed the present value of paying $3,000 per year for 5 years (at the end of each year) with a 12 percent cost of capital. Here’s the formula I used:

PV(annual_rate,Years,Amount_paid_each_year,0,0)

I named the cells D3:D5 with the labels in C3:C5 by using the Insert, Name Create command. Excel returns a net present value of -$10,814.33. (The negative sign means we are paying money out.) By omitting the last two arguments, I obtain the same answer in cell E8 with the formula 

PV(annual_rate,Years,Amount_paid_each_year)

Making payments at the end of the year is a better deal than paying out $11,000 today.

To calculate the net present value of the payments if we make payments of $3,000 on the copier for 5 years at the beginning of each year, I used the formula in cell D9:

PV(annual_rate,Years,Amount_paid_each_year,0,1)

Changing the last argument from a 0 to a 1 changes the calculation, which is now based on payments at the beginning of the year rather than payments at the end of the year. With this change, the present value of the payments is  $12,112.05, so it’s better to pay $11,000 today than to make payments at the beginning of the year. Suppose we pay $3,000 at the end of each year and we have to include an extra $500 payment at the end of year 5. We can find the present value of all our payments by including a future value of $500, using the following formula:

PV(annual_rate,Years,Amount_paid_each_year,500,0).

In this calculation, the present value of the payments equals $11,098.04. 

If I invest $2,000 a year for 40 years toward my retirement and earn 8 percent a year on my investments, how much will I have when I retire?

In this situation, we want to know the value of an annuity in future dollars (40 years from now) and not in today’s dollars. This is a job for the FV or future value function. The future value function calculates the future value of an investment assuming periodic, constant payments with a constant interest rate. The syntax of the FV function is as follows:

FV(rate,#per,[pmt],[pv],[type])

■ Rate is the interest rate per period. In our example, rate is 0.08

■ #per is the number of periods in the future at which you want to

compute the future value. #per is also the number of periods during which the annuity payment is received. In our case, #per equals 40.

■ Pmt is the payment made each period. In this example, pmt is -$2,000. The negative sign indicates we are receiving money.

■ Pv is the amount of money (in today’s dollars) owed right now. In our case, pv equals $0. If we owed someone $10,000 today, pv would equal $10,000. If we had $10,000 in the bank today, pv would equal -$10,000. If pv is omitted, it’s assumed to equal zero.

■ Type is 0 or 1, and it indicates when payments are due or money is deposited. If type equals 0 or is omitted, money is deposited at the end of a period. In our example, type is 0 or omitted. If type equals 1, payments are made or money is deposited at the beginning of a period. The file FV.xls, shown in Figure 10-2, contains the resolution to this question. In cell B7, I’ve entered the formula FV(Rate,Years,-Annual_deposit,0,0) to find that in 40 years our nest egg will be worth $518,113.04. Notice that I entered a negative value for our annual payment because a deposit can be viewed as a negative payment. In cell C7, I obtained the same answer by omitting the last two (unnecessary) arguments. The formula entered in C7 is FV(Rate,Years,-Annual_deposit). If deposits are made at the beginning of each year for 40 years, the formula entered in cell B8, which is FV(Rate,Years,-Annual_deposit,0,1), yields the value of our nest egg in 40 years, $559,562.08. Finally, suppose that in addition to investing $2,000 at the end of each of the next 40 years, we have $30,000 with which to invest initially. If we earn 8 percent per year on our investments, how much money will we have when we retire in 40 years? We can answer this question by setting pv equal to -$30,000 in the FV function. (The negative sign indicates that we have money rather than owe someone money.) In cell B9 the formula

 FV(Rate,Years,-Annual_deposit,0,0)+FV(Rate,Years,0,-30000,1)

yields a future value of $1,169,848.68. The formula FV(Rate,Years,0,-30000,1) yields the future value (in 40 years) of $30,000 received today. The formula includes type = 1 because $30,000 is received today. I used a negative sign with the $30,000 because we are “owed” -$30,000. By the way, because our money is growing at 8 percent a year, FV(Rate,Years,0,-30000,1) simply yields (1.08)40($30,000).

Figure 10-2

